ANALYTICAL BIOCHEMISTRY **181**, 163–166 (1989)

How to Build an Inexpensive Cyclotherm Instrument for Automated Polymerase Chain Reaction

Michael Collasius,* Heinrich Falk,* Christian Ciesler,† and Günter Valet*

**Mildred-Scheel-Labor für Krebszellforschung and †Abteilung für NMR-Spektroskopie, Max-Planck-Institut für Biochemie, D-8033 Martinsried, Federal Republic of Germany*

Received February 15, 1989

The Cyclotherm instrument is a functionally fully equivalent but inexpensive alternative to commercial instruments for automated polymerase chain reaction (PCR). It can be rebuilt under conditions of a biochemical laboratory for less than $1000. A Peltier element is used for heating and cooling of the reaction vials and the temperature and timing of the PCR cycles are controlled by a BASIC program in a SHARP PC 1600 low cost computer.

The PCR[1] is a method for primer-directed enzymatic, nearly exponential amplification of specific DNA sequences. It is possible to synthesize millions of copies of specific DNA fragments with two primers, complementary to the flanking regions on opposite strands of the desired DNA fragment (1–3). Repeated cycles of three different reaction temperatures are needed: the first temperature for heat-denaturing the DNA, the second for annealing, and the third for extension of the primers. In earlier reports the Klenow fragment was used to catalyze the extension step. After each denaturation step the enzyme was temperature inactivated and had to be added again to the reaction mix. The PCR is easier to handle and more specific products are obtained since thermostable DNA polymerase of *Thermus aquaticus* has become available (4,5). The PCR can be automated by using a cyclic heating and cooling device. Commercially available instruments are rather costly with prices between $5000 to over $10,000. The automated Cyclotherm was developed as an inexpensive alternative which can be easily rebuilt under laboratory conditions.

A thermoelectric Peltier element is used for the heating and cooling of the PCR assay. Heating at one side and cooling on the other side depend on a dc current passing through the interface of a Peltier element between two different metals (6). By changing the polarity of the electric current it is possible to reverse the direction of the heat.

The PCR cycle in the Cyclotherm instrument can be carried out repeatedly by using a small programmable computer which measures the reaction temperature electronically and controls the direction as well as the presence or absence of the electrical current through the Peltier element by relays or transistors.

MATERIALS AND METHODS

Hardware

The electronic circuits were built on Europa-cards and assembled in a 19-in. Compac housing (Schroff, Munich, FRG) (Fig. 1). The Peltier element was placed between the ventilator (Pabst 11W, St. Georgen, FRG)-cooled aluminium radiator fins on one side and a custom-made aluminium or copper block (45.0 × 35.0 × 15.0 mm) for 12 reaction vials of 0.5 ml on the other side. Thermal resistance between the metal interfaces was reduced with a thermal energy conducting glue. The electronic part (Table 1) of the instrument was built with no other equipment than a soldering station and a voltage/current/resistance measurement multimeter.

Temperature measurement. The temperature in the aluminium block is obtained from the voltage between the basis and the emitter of a transistor (BC 109) (Fig. 2). The voltage is temperature dependent at the constant basis-emitter current. The constant basis-emitter current is maintained by an operational amplifier which inverts and amplifies the temperature-dependent basis-emitter voltage of the transistor. The amplified basis-emitter voltage of the transistor can be adjusted to the required temperature range (0–100°C) by changing the resistance of the two trim-potentiometers R_0 and R_{Amp}. The 0°C point of the temperature scale is set with the R_0

[1] Abbreviations used: PCR, polymerase chain reaction; ADC, analog-digital convertor.

0003-2697/89 $3.00



FIG. 1. Front view of the Cyclotherm instrument.



FIG. 2. Cyclotherm temperature circuit.

potentiometer. It corresponds to an offset of the operational amplifier by 200 mV. The gain of the operational amplifier is adjusted by the potentiometer R_{Amp} to 20 mV/°C; i.e., an amplified analog signal of 2.2 V corresponds to 100°C.

Temperature control. The Sharp PC 1600 low cost computer was used for the temperature control and regulation of the Cyclotherm instrument. It can be programmed in BASIC and has a built in analog-digital converter (ADC). The ADC converts the analog signal from the operational amplifier into a digital number with a resolution of 8 bit; i.e., a range of 256 temperature steps is available. The ADC allows a maximum precision of 0.39°C per ADC step. When the temperature range is set between 0 and 100°C by the R_0 and R_{Amp} potentiometers, an accuracy of ±1°C is obtained for the temperature measurement.

Switching the Peltier thermoelement. Three lines of the serial RS-232-C interface of the PC 1600 computer were used for signal transmission to the relays: RTS (ready to send; pin 4), DTR (data terminal ready; pin 14), and signal ground (pin 7). One relay switches the current for the Peltier element on and off (Fig. 3), the other relay changes the polarity of the Peltier current. Both relays were Siemens Type DO 712-F104 (Siemens, Munich, FRG). The DC coil of the relay has a resistance of 58 Ω and draws 103 mA electrical current at 6 V. The relays are able to switch 5 A at 220 V. The electrical signals of the serial interface have to be amplified by the inverting TTL 7406 chip, because they are too weak to switch the relays directly. The inversion of the signal by the TTL 7406 has an important security aspect. In case the computer is switched off, the signal line is automatically set "high," i.e., the electrical current for the Peltier element is switched off.

A diode prevents the destruction of the TTL 7406 through reverse currents which may be induced during switching of the relays. The MI1069 Peltier element was obtained from Nucletron (Munich, FRG).

TABLE 1

List of Electronic Components[a]

Temperature circuit		Peltier circuit	
Power supply		Power supply	
Input	AC 220 V/50/60 Hz	Input	AC 220 V/50/60 Hz
Output	DC 6.5 V/600 mA	Output	DC 15 V/4 A
Voltage regulation	+5 V 7805	Voltage regulation	+5 V 7805
Capacitor	.47 µF	Capacitor 2×	10 µF
	10 µF	Diode 2×	5 V
	10 nF	IC	TTL 7406
Trim-potentiometer	5.0 kΩ	Relay 2×	DO 712-F104, Siemens
	2.2 kΩ	Peltier element	MI 1069, Nucletron
Resistors	20.0 kΩ	Ventilator	18 W, 220 V, Pabst
	7.5 kΩ		
	2.4 kΩ		
	1.0 kΩ		
Operational amplifier	CA 3140 E		
Transistor	BC 109		

[a] A more detailed part list, the addresses of suppliers and suggestions for the mechanical setup are available on request from the authors.



FIG. 3. Cyclotherm Peltier circuit.

Power supply. The Cyclotherm instrument is driven by two power supplies. One is for the Peltier element which is operated at optimum efficiency (48 W) with 4 A DC current at 14 V. The second power supply runs the temperature circuit and the computer.

The temperature circuit is galvanically separated from the Peltier circuit to prevent electronic interference from the current switching relays.

Software

The temperature regulation program is written in BASIC (Table 2). The default values for the cycle temperature TP(1)–TP(5) and time TI(1)–TI(5) are stored in line 80. Time is given in the format "MMSS" where MM are minutes and SS are seconds (line 110). Temperature is given in degree Celsius (line 100). The operation parameters can be changed for each run by answering the question "Correct?" (line 130) with "N" for No to obtain a program stop (line 150). New values can be entered for the temperature, the time, and the number of cycles by typing variable name and value; i.e., T3 = 40 will set the extension temperature to 40°C. By typing CONT the program will resume. The program executes the denaturation, annealing, and extension phase of the PCR by sequentially jumping (lines 180–240) into the subroutines: change temperature (lines 290–430) or hold temperature (lines 450–610). The analog signal of the temperature circuit is digitized by the ADC upon execution of the input command "AIN" (lines 310–350, 470–510) and converted into a temperature value in degrees Celsius by the program (lines 360, 520). Depending on the measured temperature the Peltier element is set to cooling (lines 400, 570), heating (lines 410, 580), or switched on/off (line 590 L:) by the command OUTSTAT "COM1:".

The program can be modified to obtain more sophisticated action if desired or it can be transferred to another computer from which the Cyclotherm instrument may be operated by replacing the hardware specific commands AIN and OUTSTAT"COM1:" with appropriate commands.

Polymerase Chain Reaction

The polymerase chain reaction was carried out in a final volume of 100 μl. Reaction buffer (final concentration 2.0 mM $MgCl_2$, 50 mM KCl, 10 mM Tris–HCl, pH 8.3) and dNTPs (dATP, dCTP, dGTP, and TTP each 200 μM) were mixed. Two 21-mer primers (P1.2, 5′AATAGAGTTAGGCAGGGATAT3′; P2.2, 5′GATAAGTGCTAAGGATCCGTT3′) complementary to a part of the HIV-1 env gene were added in a concentration of 14 pM. The reaction was started with 2.5 units of Taq polymerase, following addition of the templates of HIV-1 (1 fmol) and phage λ (100 ng).

TABLE 2

The BASIC Program of the Cyclotherm Instrument

```
10 ARUN
20 REM  ** PCR PROGRAM *****************
30 ERASE TP(),TI()
40 DIM TP(5),TI(5)
50 DT=2:C=0
60 SETDEV "COM1:"
70 OUTSTAT "COM1:",0
80 DATA 94,0130,37,0230,72,0330,94,0130,04,-1,25
90 READ TP(1),TI(1),TP(2),TI(2),TP(3),TI(3),TP(4),TI(4),TP(5),TI(5),CY
100 PRINT USING "####";"Temp";TP(1);TP(2);TP(3);TP(4);TP(5)
110 PRINT USING "####";"Time";TI(1);TI(2);TI(3);TI(4);TI(5)
120 PRINT USING "####";"Cycles ";CY
130 INPUT "Correct ?";A$;
140 IF INSTR (A$,"Y")>0THEN 180
150 STOP
160 GOTO 100
170 REM  ** MAINPROGAM *****************
180 TE=TP(1):ZE=TI(1):GOSUB 280:'**Denaturing
190 FOR C=1TO CY
200 TE=TP(2):ZE=TI(2):GOSUB 280:'**Annealing
210 TE=TP(3):ZE=TI(3):GOSUB 280:'**Extension
220 TE=TP(4):ZE=TI(4):GOSUB 280:'**Denaturing
230 NEXT C
240 TE=TP(5):ZE=TI(5):GOSUB 280:'**Last step
250 OUTSTAT "COM1:",0
260 END
270 REM  ** SUBROUTINES *****************
280 REM  ** CHANGE TEMP *****************
290 CLS
300 AE=(TE*2.18)+17
310 AD=AIN :' ** READ ADC
320 FOR WIED=1TO 10
330 AD=AD+AIN
340 NEXT WIED
350 AT=AD/WIED
360 T=INT ((AT-17)/2.18)
370 CURSOR 0,0
380 PRINT USING "###";"Temp ";T;CHR$ (&F8);"C SetTemp";TE;CHR$ (&F8);"C"
390 PRINT "Cycle ";C;" of ";CY
400 IF AE<(AT-DT)THEN OUTSTAT "COM1:",2:' ** COOLING
410 IF AE>(AT+DT)THEN OUTSTAT "COM1:",3:' ** HEATING
420 IF ABS (AE-AT)<DTTHEN GOTO 440
430 GOTO 310
440 REM  ** HOLD TEMP *****************
450 TIME$ ="00:00:00"
460 T0=TIME
470 AD=AIN :' ** READ ADC
480 FOR WIED=1TO 10
490 AD=AD+AIN
500 NEXT WIED
510 AT=AD/WIED
520 T=INT ((AT-17)/2.18)
530 CURSOR 0,0
540 PRINT USING "###";"Temp ";T;CHR$ (&F8);"C Set Temp";TE;CHR$ (&F8);"C"
550 PRINT "Cycle ";C;" of ";CY
560 PRINT "TIME ";RIGHT$ (TIME$ ,5)
570 IF AE<(AT-DT)THEN OUTSTAT "COM1:",2:' ** COOLING
580 IF AE>(AT+DT)THEN OUTSTAT "COM1:",3:' ** HEATING
590 IF ABS (AE-AT)<DTTHEN OUTSTAT "COM1:",0:' ** PELTIER CURRENT OFF
600 IF ZE<0THEN GOTO 470ELSE IF (TIME -T0)>=(ZE/10000)THEN RETURN
610 GOTO 470
```

[a]A more comfortable program is available on request from the authors.

RESULTS

Cyclotherm was tested with a typical PCR cycle profile of: 2 min denaturing at 94°C; 2.5 min annealing at 37°C; 3.5 min extension at 72°C. The cooling rate was 0.6°C s^{-1}. During heating the temperature increased with a rate of 0.8°C s^{-1}. To assert whether Cyclotherm is suitable for PCR a whole run of 30 cycles was made. No temperature difference was observed between individual cycles as monitored on an analog voltage recorder (Fig. 4). A 160-bp product was PCR amplified (Fig. 5, lane 2), using the HIV-1 cDNA template and two primers, complementary to the env gene of HIV-1.

DISCUSSION

Our objective with Cyclotherm was to build a simple and inexpensive instrument for PCR without compromise to performance. The Cyclotherm instrument is controlled by the low cost Sharp PC 1600 computer. It contains an ADC for analog input and a serial interface for data output.

Alterations of the temperature protocol can be easily accomplished by the user, due to the flexibility of BASIC programming. The software printout (Table 1) is a minimum version to carry out a full PCR. Increasing the number of temperature changes per cycle is achieved by entering new time (TI_n) and temperature (TP_n) variables into the program and an additional GOSUB statement into the FOR-NEXT loop (lines 190–230). Furthermore, the slope of temperature increase/decrease may be modified by switching the on/off relay in a time-dependent manner (lines 400–430).

Cyclotherm has an active cooling system in contrast to some other commercially available systems. Therefore the reaction mixtures can be cooled below room temperature. This is an advantage when the PCR is running overnight and the reaction product must be kept cool following completion of the reaction.

Cyclotherm can be expanded by using additional Peltier elements either together with a larger aluminium block suitable for more vials or for faster heating and cooling of the reaction assays in the present block. The relays can be replaced by current switching transistors.

The Cyclotherm is inexpensive, but functionally fully equivalent (Figs. 4 and 5) to commercially available automated PCR instruments, all of which are about 5 to 10 times more expensive.



FIG. 4. A typical PCR temperature profile obtained with the Cyclotherm instrument.



FIG. 5. PCR amplified 160-bp fragment of the HIV-1 env gene after 30 temperature cycles in the Cyclotherm instrument. A 123-bp weight standard (lane 1) and the amplified template of 160 bp (lane 2). The other lanes show control mixes: (lane 3) without template, (lane 4) without primers, and (lane 5) with λ DNA, which was not amplified by these primers. The bands were separated on an ethidium–bromide-stained 8% polyacrylamide gel. For details of the PCR assay see Materials and Methods.

ACKNOWLEDGMENTS

We gratefully acknowledge the technical assistance of P. Hummel and E. Glossner.

REFERENCES

1. Saiki, R. K., Scharf, S., Faloona, F., Mullis, K. B., Horn, G. T., Erlich, H. A., and Arnheim, N. (1985) *Science* **230,** 1350–1354.
2. Mullis, K. B., and Faloona, F. (1987) *in* Methods in Enzymology (Wu, R., Ed.), Vol. 155, pp. 335–350, Academic Press, San Diego, CA.
3. Higuchi, R., Beroldingen v., C. H., Sensabaugh, G. F., and Erlich, H. A., (1988) *Nature (London)* **332,** 543–546.
4. Chien, A., Edgar, D. B., and Trela, J. M. (1976) *J. Bacteriol.* **127,** 1551–1557.
5. Erlich, H. A., Gelfand, D. H., and Saiki, R. K. (1988) *Nature (London)* **333,** 461–462.
6. Gerthsen, Ch., Kneser, H. O., and Vogel, H. (1982) Physik, 13th ed., p. 295, Springer-Verlag, Berlin/Heidelberg/New York.